Frequency Responses in the Reflection Coefficient Plane Plotted by Digital Computer

It is very often necessary to predict changes in either reflection coefficient ρ, or VSWR, or input immittance as a function of frequency, say, caused by adjustment of some variable in the load of a transmission-line, or its transformation towards the generator. For an active load, e.g. a tunnel-diode amplifier, information on gain and stability may be required. The task is generally tedious if done manually.

In this correspondence details are given of a computer program which provides a plot of active or passive real frequency immittances as on a Smith chart, utilizing the complete ρ-plane. It takes the form of a subroutine into which impedance or admittance values are fed as calculated by the main program and which generates a print-out of voltage or current reflection coefficient, respectively.[1] It is designed both as an extension of a previous paper,[2] and for application to transmission-line problems generally.

Since a main program is needed to read the immittance values or to calculate them, microwave engineers unfamiliar with programming should consult their programming advisers. Minor changes to suit user or machine may be found necessary. The straightforward nature of the program obviates the usual need for a flow diagram. The specific example shown in Fig. 1 is written in FORTRAN IV and is briefly described as follows.

The first card indicates the name by which the subroutine is called, i.e. CALL CHART 1. The complete COMMON statement as shown, and a COMPLEX statement containing FUNC (201), must appear in the main program. FUNC stands for impedance or admittance according to whether ITYPE equals 1 or 2, and must be normalized in the main program. ITYPE also governs the page heading of the chart appropriately. NUM corresponds to the number of points entering the subroutine from the main program, in which both it and ITYPE are explicitly set before the subroutine is called. If these conditions are not satisfied the program faults.[3] In the present example, the

Manuscript received April 25, 1966; revised May 16, 1966. This work was carried out while the author held a Research Studentship from the Science Research Council, London, England.

[1] Two basic methods of storing and plotting were considered. The first was to represent a square area on the page by a matrix. The points could be entered in any order, the printing instructions ensuring the correct sequence. The second was to define a row matrix containing only the relevant points. This method would necessitate sorting the points into the print-out sequence before the printing instructions could be given. For a considerable number of points, the coordinates of which were not explicitly related or controlled, the first method would tend to use more storage space in the computer, but take less time than the second. The IBM 7090 was available to the author. As this machine is not time-sharing, it was thought that the first method would be preferable.

[2] J. W. Bandler, "Stability and gain prediction of microwave tunnel-diode reflection amplifiers," *IEEE Trans. on Microwave Theory and Techniques*, vol. MTT-13, pp. 814–819, November 1965.

[3] See statement numbers 40 and 45 in Fig. 1. Failure to observe the correspondence of NUM to the number of points may result in missing points or erroneous ones.

```
      SUBROUTINE CHART1
      COMMON /RPLANE/FUNC,ITYPE,NUM
      COMPLEX FUNC(201),RHO
      DIMENSION PNT(57,95),C(10)
      DATA (C(I),I=1,10)/1H ,1H-,1HX,1HO,1H+,1HP,1HR,1H*,1HN,1HS/
      IF(NUM.GT.200) GO TO 15
      IF(ITYPE.EQ.1) GO TO 1
      IF(ITYPE.EQ.2) GO TO 2
      GO TO 16
    1 WRITE(6,20)
      GO TO 3
    2 WRITE(6,25)
    3 NUM=NUM+1
      FUNC(NUM)=FUNC(1)
      DO 4 M=1,57
      DO 4 N=1,95
    4 PNT(M,N)=C(1)
      DO 5 N=2,94
    5 PNT(29,N)=C(2)
      PNT(29,48)=C(3)
      PNT(29,1)=C(4)
      PNT(29,95)=C(4)
      J=NUM-2
      DO 9 I=2,NUM
      RHO=(FUNC(I)-1.)/(FUNC(I)+1.)
      GAIN=CABS(RHO)
      IF(GAIN.GT.1.) GO TO 8
      N=48.5+47.*REAL(RHO)
      M=29.5-28.*AIMAG(RHO)
      PNT(M,N)=C(5)
      IF(I.LT.J) GO TO 9
      IF(I.LT.NUM) PNT(M,N)=C(6)
      IF(I.EQ.NUM) PNT(M,N)=C(7)
      GO TO 9
    8 RHO=1./CONJG(RHO)
      N=48.5+47.*REAL(RHO)
      M=29.5-28.*AIMAG(RHO)
      PNT(M,N)=C(8)
      IF(I.LT.J) GO TO 9
      IF(I.LT.NUM) PNT(M,N)=C(9)
      IF(I.EQ.NUM) PNT(M,N)=C(10)
    9 CONTINUE
      WRITE(6,30) ((PNT(M,N),N=1,95),M=1,19)
      WRITE(6,35) ((PNT(M,N),N=1,95),M=20,57)
      GO TO 100
   15 WRITE(6,40)
      GO TO 100
   16 WRITF(6,45)
  100 RETURN

   20 FORMAT(17H1IMPEDANCE CHART , 97X,17H  DB   RHO    VSWR )
   25 FORMAT(17H1ADMITTANCE CHART, 97X,17H  DB   RHO    VSWR )
   30 FORMAT                 (19X,95A1,17H   0   1.0    INF  /              1
     1      19H RHO-PLANE         ,95A1,17H             55.0  /              2
     2                        19X,95A1,17H             27.0  /              3
     3                        19X,95A1,17H   1    .9   17.7  /              4
     4                        19X,95A1,17H             13.0  /              5
     5                        19X,95A1,17H             10.2  /              6
     6                        19X,95A1,17H   2    .8    8.33 /              7
     7                        19X,95A1,17H              7.00 /              8
     8                        19X,95A1,17H   3    .7    6.00 /              9
     9                        19X,95A1,17H              5.22 /             10
     1                        19X,95A1,17H   4          4.60 /             11
     2                        19X,95A1,17H        .6    4.09 /             12
     3                        19X,95A1,17H   5          3.67 /             13
     4                        19X,95A1,17H              3.31 /             14
     5                        19X,95A1,17H   6    .5    3.00 /             15
     6                        19X,95A1,17H              2.73 /             16
     7                        19X,95A1,17H              2.50 /             17
     8                        19X,95A1,17H   8    .4    2.29 /             18
     9      19H POSITIVE IMAGINARY,95A1,17H              2.11 )             19
   35 FORMAT                 (19X,95A1,17H  10          1.95 /             20
     1                        19X,95A1,17H        .3    1.80 /             21
     2                        19X,95A1,17H  12          1.67 /             22
     3                        19X,95A1,17H        .2    1.55 /             23
     4                        19X,95A1,17H  15          1.44 /             24
     5                        19X,95A1,17H  17          1.33 /             25
     6                        19X,95A1,17H  20    .1    1.24 /             26
     7                        19X,95A1,17H  23          1.15 /             27
     8                        19X,95A1,17H  29          1.07 /             28
     9                        19X,95A1,17H0 INF   .0    1.00 ,             29
     1                   9(/19X,95A1)/                                     38
     2      19H NEGATIVE IMAGINARY,95A1 ,                                  39
     3                  12(/19X,95A1)/                                     51
     4      19H J W BANDLER       ,95A1,12H   FREQUENCY/                   52
     5      19H MICROWAVE LAB     ,95A1,12H   RESPONSE /                   53
     6      19H ELEC ENG DEPT     ,95A1 /                                  54
     7      19H IMPERIAL COLLEGE  ,95A1 /                                  55
     8      19H LONDON SW 7       ,95A1 /                                  56
     9      19H SEPT 1965         ,95A1 )                                  57
   40 FORMAT(47H1THE MAXIMUM NUMBER OF POINTS HAS BEEN EXCEEDED)
   45 FORMAT(36H1THE CHART TYPE HAS NOT BEEN DEFINED)
      END
```

Fig. 1. The complete subroutine program in FORTRAN IV.



Fig. 2. Part of tunnel-diode amplifier.
Example
Tunnel-diode: $R=45\Omega$, $C=2.0$pF, $r=1.0\Omega$,
$L=0.12$nH, C(package) $=0.3$pF.
Stabilizing filter: $Y_1=35.425$mmhos, $l_1=2.4175$cm,
$Y_2=52.775$mmhos, $l_2=2.7762$cm,
$Y_3=52.495$mmhos, $l_3=2.5844$cm.
Bias resistance: 40Ω

maximum square area which would fit the available page was used.[4]

The – sign is used for the center-line (zero reactance or susceptance), with X at the center-point (ρ-plane origin). O is printed at either extremity, with an extra O on the right to indicate infinite immittance (see Fig. 3). The +, P, and R are used for immittances with positive real parts, the *, N, and S for those with negative real parts. R or S characterize the first point (arrow tail), P or N the last two points (arrow head).[5]

When the real part of the immittance is negative, the response is reflected back into the unit circle by the transformation[2]

$$\rho' = \frac{1}{\rho^*}$$

where ρ^* is the complex conjugate of ρ, accompanied by the appropriate change in the character representing the point.

After all the points have been assembled, they are plotted on a new page with the



DB	RHO	VSWR
0	1.0	INF
		55.0
		27.0
1	.9	17.7
		13.0
		10.2
2	.8	8.33
		7.00
3	.7	6.00
		5.22
4		4.60
	.6	4.09
5		3.67
		3.31
6	.5	3.00
		2.73
		2.50
8	.4	2.29
		2.11
10		1.95
	.3	1.80
12		1.67
	.2	1.55
15		1.44
17		1.33
20	.1	1.24
23		1.15
29		1.07
INF	.0	1.00

Fig. 3. Smith chart plot of Z_{IN} of Fig. 2 normalized to 10Ω. Frequencies: 0 to 175Mc/s in steps of 25Mc/s; 200Mc/s to 4.6Gc/s in steps of 100Mc/s. Quantization inevitably caused some overwritten points; hence, they are missing.

scales, as shown in Fig. 3. Note that the numbers constituting the scales are taken to represent the values at the middle of the lines. The plot shown[6] is a frequency response of the input impedance Z_{IN} of the network of Fig. 2, for the parameter values shown. This network includes the stabilizing and biasing circuit of an S-band rectangular waveguide tunnel-diode amplifier, for which, for example, a rapid evaluation of the effects of diode parameter variation on stability, as indicated by Nyquist-type plots,[2] is essential.

The program is readily modified to suit alternative scales, sizes, layouts, etc. An expanded chart could be realized by dividing RHO in the program by the maximum magnitude to be displayed, and then restricting excursive points to the circumference, or eliminating them altogether. The adjoining scales must, for this case, be altered or removed. The program could also be rewritten to accommodate automatic scaling which would be determined internally or externally to the subroutine, and possibly generating a print-out of the scales. The chart size can similarly be altered. For some applications a specially prepared transparency of a Smith chart may be used in conjunction with the computer plot to obtain associated parameters more accurately, e.g. the phase of ρ.

It is felt, however, that the whole ρ-plane merits a subroutine of its own, such as the one described here, as it is so frequently used.

J. W. Bandler
Dept. of Elec. Engrg.
Imperial College
London, England

[4] This area is provided by a 57 by 95 matrix. The DIMENSION statement declares the size of the matrix, and also the number of characters that can be plotted, here 10. The DATA statement defines these characters and groups them in the order of importance with the least important, a blank, first. Initially, all the elements of the matrix are made blanks to clear it of previous points (DO loop terminating on statement 4).

Clearly, the co-ordinates can only be located in the matrix at discrete positions. To effect this, use is made of the fact that a real variable entering an integer register is truncated at the decimal point. Hence, each point is located in row M, column N by

$$M = \frac{A+2.}{2.} - \frac{A-1.}{2.}\,\mathrm{Im}(\rho)$$

$$N = \frac{B+2.}{2.} + \frac{B-1.}{2.}\,\mathrm{Re}(\rho)$$

where the matrix has A rows and B columns, both of which must be odd to ensure symmetry of the chart about both real and imaginary axes.

[5] The sequence of operations ensures that $C(5)$ to $C(10)$ can overwrite $C(1)$ to $C(4)$; P, R, N, or S can never be overwritten by + or *; R or S can overwrite all other characters (caused by operating on FUNC(1) last; hence, the possibility of obtaining FUNC(201) due to NUM being increased by 1).

[6] The scales and labelling of Fig. 3 have been rearranged here to facilitate a more convenient printing layout. The entire square area within which the chart is contained is, in fact, free of labelling, and the scales are printed vertically from the center-line upwards, alongside the square.